XMLのテキストノードを要素ノードでラップしたい（条件あり）


以下のようなXMLサンプルがあるとき、XSLでHTMLタグに変換したいと思っています。

[code]

    <root>
        <section>
            <container>
                aaa
                <box>
                    book
                </box>
                bbb
                <box>
                    pen
                </box>
                ccc
                <superscript>
                    3
                </superscript>
                ddd
            </container>
        </section>
    </root>
    
[/code]

XSLで、以下のような結果を得ることは出来るでしょうか？「aaa」「bbb」「ccc3ddd」をpタグでラップし、「box」にはdivタグを、「superscript」にはspanタグを割り当てたいと考えています。よろしければ、アドバイスお願いします。

[code]

    <div>
        <p>aaa</p>
        <div>book</div>
        <p>bbb</p>
        <div>pen</div>
        <p>ccc<span>3</span>ddd</p>
    </div>
    
[/code]




今までいろいろXSLTで変換をやっていますが、入力XMLでもう少しちゃんとタグ付けをしていてくれたら良いのにと思ってもなかなかお客様の元データを直すわけにはいかず、XSLTスタイルイート側で吸収しなければならないことが多々ありました．

今回のご質問も"aaa","bbb","ccc"~"ddd"がpでタグ付けされていれば電車道ですね．でもそう行かないのでしょう．このような場合グルーピングの問題として解決することが出来るのではないか？と思います．つまりcontainer要素以下のnode()をbox要素かそうでないものかによってグルーピングするのです．

  * box要素の場合は個々にdiv要素に変換します．
  * 他のノードの場合はp要素にグルーピングして変換します

argusさんのスタイルシートを参考にさせていただき、XSLT 2.0のxsl:for-each-groupで実現したのが以下のコードです．

[code]

    <?xml version="1.0" encoding="UTF-8"?>
    <xsl:stylesheet xmlns:xsl="http://www.w3.org/1999/XSL/Transform"
        xmlns:xs="http://www.w3.org/2001/XMLSchema"
        exclude-result-prefixes="xs"
        version="2.0">
        <xsl:output method="html" encoding="UTF-8"/>
    
        <xsl:template match="/">
            <html>
                <body>
                    <xsl:apply-templates/>
                </body>
            </html>
        </xsl:template>
    
        <xsl:template match="section">
            <xsl:for-each select="container">
                <div>
                    <xsl:for-each-group select="node()" group-adjacent="name(.) eq 'box'">
                        <xsl:choose>
                            <xsl:when test="current-group()[self::box]">
                                <xsl:apply-templates select="current-group()"/>
                            </xsl:when>
                            <xsl:otherwise>
                                <p>
                                    <xsl:apply-templates select="current-group()"/>
                                </p>
                            </xsl:otherwise>
                        </xsl:choose>
                    </xsl:for-each-group>
                </div>
            </xsl:for-each>
        </xsl:template>
    
        <xsl:template match="box">
            <div>
                <xsl:apply-templates/>
            </div>
        </xsl:template>
    
        <xsl:template match="superscript">
            <span>
                <xsl:apply-templates/>
            </span>
        </xsl:template>
    
        <xsl:template match="text()">
            <xsl:value-of select="normalize-space(.)"/>
        </xsl:template>
    
    </xsl:stylesheet>
    
[/code]

結果は以下のようになります．

[code]

    <html>
       <body>
          <div>
             <p>aaa</p>
             <div>book</div>
             <p>bbb</p>
             <div>pen</div>
             <p>ccc<span>3</span>ddd
             </p>
          </div>
       </body>
    </html>
    
[/code]

それなりに自然な流れのスタイルシートになっていると思います．（何故か"ddd"のあとで改行されていてそれだけ疑問ですが？？？）

参考にしていただければ幸いです．

